' Calculate Income 2 - © 1993-2002 Aiman Hanna

' This program illustrates Function Procedures.

' Functions are similar to subs with one important difference,

' which is that Functions can return values. When calling (executing)

' a function and getting the returned value, the parentheses are required.

' In general, if you need to use the value that will be calculated by the procedure

' after the procedure terminates, then you must use functions instead of subs

' Key Points:

'       1) Function Procedures

'       2) Returning a value to the caller

'       3) Using declaration characters with functions

Option Explicit

Private Sub cmdCalcIncome_Click()

    ' This sub will get the needed information to calculate

    ' an yearly income. The sub will then call one function to calculate

    ' the income. The sub will then call another function to calculate the taxes

    ' based on the returned income value.

    Dim hours As Single, rate As Currency

    Dim income As Currency, taxes As Currency

    Const minhours As Integer = 1, minrate As Currency = 7.25

    ' Get number of hours and rate from user and validate them

    ' The minimum number of hours is assumed to be 1. The minimum

    ' rate is assumed to be 7.25$

    ' Make sure first that the user enters the two needed values

    ' If any of the values is missing, terminate the execution of this sub

    If txtHours.Text = "" Or txtRate.Text = "" Then

        Call MsgBox("You must enter the two needed values.", vbOKOnly, "Missing Value(s)")

        Exit Sub

    End If

    ' Now the two values are entered, read them and then perform the needed validation

    hours = txtHours.Text

    rate = txtRate.Text

    ' Validate the rate

    While rate < minrate

        Call MsgBox("Invalid value of hourly rate. Min is " & minrate & ".", vbOKOnly, "Invalid Hours")

        rate = InputBox("Enter the hourly rate", "Rate")

        If rate >= minrate Then   ' user enters a good value, update the rate textbox

            txtRate.Text = rate

        End If

    Wend

    ' Now validate the hours

    While hours < minhours

        Call MsgBox("Invalid number of hours per week. Min is " & minhours & ".", vbOKOnly, "Invalid Hours")

        hours = InputBox("Enter the number of working hours per week", "Hours")

        If hours >= minhours Then   ' user enters a good value, update the hours textbox

            txtHours.Text = hours

        End If

    Wend

    ' At this point we have valid values. Call the ComputeIncome sub

    ' to display the yearly salary

     income = ComputeIncome(hours, rate)

     lblIncomeMsg.Caption = "The yearly salary is " & Format$(income, "Currency") & "."

     ' Now pass this income to the TaxCompute function to compute the taxes

     taxes = TaxCompute(income)

     lblTaxMsg.Caption = "The taxes for this amount is " & Format$(taxes, "Currency") & "."

     ' finally show the net income

     lblNetIncomeMsg.Caption = "The net income is " & Format$(income - taxes, "Currency") & "."

End Sub

Private Function ComputeIncome(hrs As Single, rt As Currency) As Currency

    Const weeksperyear As Integer = 52

    ComputeIncome = hrs * rt * weeksperyear     ' this value is what is returned

End Function

Private Function TaxCompute@(inc As Currency)  ' this is the same as: Private Function TaxCompute() As Currency

    ' There are three different tax brackets depending on the salary

    Const tb1 As Single = 0.25, tb2 As Single = 0.35, tb3 As Single = 0.5

    ' Notice that it would be better to have these three salary ranges as constants as well

    ' for simplicity (not to lose focus on the discussed point), we will just use the

    ' values themselves

    If inc < 27000 Then

        TaxCompute = inc * tb1

    ElseIf inc < 49000 Then

        TaxCompute = inc * tb2

    Else

        TaxCompute = inc * tb3

    End If

End Function

Private Sub cmdExit_Click()

    End

End Sub

